On page 892 we began our investigation of geometric patterns in LOGO by generating cycloids — the curves traced by circles rolling on a straight line; this by no means exhausts the possibilities of the moving circle, however. Instead of moving along a straight line, the generating circle could rotate within another circle, and the path the drawing point traces is called a *hypercycloid*.

Much of the problem remains the same — we still need to move the centre of the generating circle and then move out to the drawing point on the circumference. However, now we need to keep track of two angle steps. One (HEADSTEP) is for working out the path of the centre of the generating circle, and the other (ANGLESTEP) keeps track of the heading of the drawing point with respect to the centre of this circle. As the centre of the smaller circle rotates, the drawing point rotates in the opposite direction. It is possible to show that the sizes of the two angles are related by the formula:

```
ANGLESTEP = HEADSTEP X
( RADIUS1 / RADIUS2 - 1)
```

where RADIUS1 is the radius of the fixed circle, and RADIUS2 that of the rotating one. The procedure HYPERCYCLOID takes RADIUS2 as an input, thus allowing us to trace out a number of different hypercycloids.

```
TO HYPERCYCLOID :RADIUS2
   SETSCREEN
   MAKE "PI 3.14
   MAKE "RADIUS1 60
   MAKE "DIFFERENCE :RADIUS1 -
   :RADIUS2
   MAKE "HEADSTEP 6
   MAKE "CIRCUMFERENCE 2 * :PI *
   :DIFFERENCE
   MAKE "STEP :CIRCUMFERENCE / ( 360
   / :HEADSTEP )
   MAKE "ANGLESTEP :HEADSTEP * (
   :RADIUS1 / :RADIUS2 - 1 )
   MAKE "CENTRE LIST 0 :DIFFERENCE
   MAKE "HEAD 0
   MAKE "X CENT 0
   MAKE "OLDPOS LIST :XCENT :RADIUS1
   HCYC 0
END

TO HCYC :ANG
   MOVECENTRE2
   SETPOS POS
   SETH :ANG
   FORWARD :RADIUS2
   MAKE "NEWPOS POS
   JOIN :OLDPOS :NEWPOS
   MAKE "OLDPOS :NEWPOS
   HCYC :ANG - :ANGLESTEP
END

TO MOVECENTRE2
   SETXY 0 0
   SETH :HEAD
   FORWARD :DIFFERENCE
   MAKE "CENTRE POS
   MAKE "HEAD :HEAD + :HEADSTEP
END
```

There is an interesting special case: if the radius of the rolling circle is one half that of the fixed circle then the hypercycloid becomes a straight line! In this way, motion within a circle is transformed into motion along a straight line. You might like to modify the procedures to find out what happens if the point is inside the circle, or outside the circle.

'Curve stitching' is another way of developing some interesting shapes from circles. Take two concentric circles and mark each of them out into a large number of equal arcs — say 120. Number the points, and then join them, one at a time, to points on the other circle according to some simple rule — for example, x 'maps onto' 2x. The results can be very surprising.

This activity can actually be done with needle and thread, so it's often called *curve stitching*. It can also be done with pen and paper, but obviously we would prefer you to use LOGO. Here is our version of a curve stitching program:

```
TO SETUP
   MAKE "RADIUSA 80
   MAKE "RADIUSB 60
   DRAW
   HT
   PENUP
   DRAWIT 0 0
END

TO DRAWIT :A :B
   IF :A = 120 THEN STOP
   JOIN PTA :A PTB :B
   MAKE "A :A + 1
   MAKE "B 2 * :A
   DRAWIT :A :B
END

TO PTA :NO
   SETXY 0 0
   SETH :NO * 3
   FORWARD :RADIUSA
   OUTPUT POS
END

TO PTB :NO
   SETXY 0 0
   SETH :NO * 3
   FORWARD :RADIUSB
   OUTPUT POS
END
```

You may like to investigate the patterns generated by other rules, such as x → 3x, x → 4x, etc.